
ABSTRACT
Epidemiological studies sometimes are designed to compare two groups using one-to-one matching in order
to create a control group that is comparable to a case group on one or more risk factors. Producing equal-
sized case and control samples that are matched on multiple variables can present a challenge to any
programmer. Using the PROC SURVEYSELECT procedure in SAS Version 8 ® software can facilitate the
development of such sampling programs by eliminating the need for macros. This procedure has the ability
to select samples within independent stratified subgroups. This paper presents a program that can be simply
adapted that will select a matched sample of equal numbers of cases and controls when matching on
multiple variables using simple random sampling without replacement.

INTRODUCTION
Epidemiological studies can be designed to eliminate potential confounders by using one-to-one matching to
compare two groups.  The matching reduces the numbers of covariates needed for statistical analysis, thus
simplifying and strengthening the modeling process.  Creating a random sample of equal numbers of cases
and controls, which are matched on multiple variables, can be a challenge to any SAS programmer. The
difficulty comes into accounting for the different sample sizes within each stratum created by the matching.
While this has been previously accomplished with macros and loops, the SURVEYSELECT procedure in
SAS V8 now simplifies this task.  This paper presents a simple program that can be modified by other users
to generate an analytic data set of matched cases and controls using simple random sampling without
replacement to pull independent samples stratified on one or more matched criteria.

DATA SETUP
In this example, a health plan wants to conduct a survey of diabetics who have received counseling (cases)
and a random sample of diabetics who did not (controls). The controls will be randomly matched to cases
based on the type of medical benefit they have, the age of the patient, whether or not they are on insulin,
and the anti-diabetic drug they use. For this example, we will establish a pool of eligible cases and a pool of
eligible controls and the look-up table of sample sizes to pull equal numbers from each subgroup based on
their matching criteria.  

To produce a matched random sample, PROC SURVEYSELECT requires two input data sets. The first is
the pool of eligible units from which to draw the sample. The second is a lookup table of sample sizes
associated with each matched stratum.

THE POOL OF ELIGIBLE UNITS
First, we create two data sets: one for eligible cases and one for eligible controls. Each data set contains the
variables, INDEX and SDYID. The variable INDEX is a concatenation of the four matching variables,
BENEFIT, AGE, INSULIN, and DRUG that are formatted as described below.  The variable SDYID is a
unique patient identifier. The following code creates the control data set and its new INDEX variable (similar
code can be used for cases).

/*SDYID = Patient Identifier (any format) */
/*BENEFIT = MEDICAL BENEFIT (any 4 digits - Character format) */
/*AGE = AGE OF PATIENT (2 digit number -character format) */
/*INSULIN = INSULIN YES OR NO  (0 or 1- character format) */
/*DRUG = ORAL HYPOGLYCEMIC DRUG  (2 digit number- character format)*/

DATA CTRL (KEEP=SDYID INDEX);
   SET CTRLPOOL (KEEP=SDYID BENEFIT AGE INSULIN DRUG);
INDEX = BENEFIT||AGE||INSULIN||DRUG;

THE SAMPLESIZE LOOK-UP TABLE 
PROC SURVEYSELECT eliminates the need for macros or data step loops to select a sample the data for
each index. By using a lookup table of sample sizes for each Index, PROC SURVEYSELECT will stratify its
sampling. The parameters for the sample size include 1) it must be a positive integer, 2) it cannot be greater
than the number of units in that stratum for methods using selection without replacement.  Simply compute
the number of units within each index and take the smallest number as the sample size for the index. The
look-up SAS data set should include a stratification variable (ours is named INDEX) and the sample size
variable that must be named _NSIZE_. For our purposes, we compute two output data sets (CTRLCNT and
CASECNT) illustrated using the following code for controls. Data look like the output in Figures 1 and 2,
respectively.

PROC FREQ DATA= CTRL NOPRINT;
TABLES INDEX/LIST MISSING OUT=CTRLCNT (KEEP=INDEX COUNT
RENAME=(COUNT=CTRLCNT));

Figure 1. Data set WORK.CTRLCNT

Obs     INDEX CTRLCNT 

 1     Bx1025008        2       
 2     Bx1037008        1       
 3     Cy2043008        5       
 4     Cy2043108        3      
 5     Ggg164008        1  
  

Figure 2. Data set WORK.CASECNT

Obs     INDEX       CASECNT    

 1     Bx1025008       2         
 2     Bx1037008       2         
 3     Cy2043008       9         
 4     Cy2043108       2         
 5     Ggg164008       1         
 6     Zgx164008       5*
*Cases omitted from final data set due to no controls with the index.

Merge the distributions together by the Index and determine the number of cases and controls to select for
each stratum. The sample size data set that we will feed into the PROC SURVEYSELECT statement is in
figure 3, WORK.ALLCOUNT. Note that the only necessary variables for this data set are INDEX and
_NSIZE_. The code below selects the minimum number of patients from either the cases or controls and the
last line deletes strata where there are missing cases or controls. Thus, the stratum "Zgx164008", Fig 2 obs
6, would be deleted from the ALLCOUNT data set because there are no controls with that particular index,
effectively omitting 5 cases from the final data set.

DATA ALLCOUNT; 
     MERGE CASECNT (IN=A) CTRLCNT (IN=B);
     BY INDEX;
     IF CASECNT > 0;
     IF A AND NOT B THEN CTRLCNT = 0;
     _NSIZE_ = MIN(CASECNT,CTRLCNT);
     IF _NSIZE_ GT 0;

Figure 3. Data set WORK.ALLCOUNT

Obs        INDEX     CASECNT  CTRLCNT _NSIZE_

 1     Bx1025008       2         2       2
 2     Bx1037008       2         1       1
 3     Cy2043008       9         5       5
 4     Cy2043108       2         3       2
 5     Ggg164008       1         1       1 

LIMIT THE SELECTION POOLS
We limit the pools of cases and controls to those in strata with non-zero sample sizes from
WORK.ALLCOUNT.  This step is not necessary, but can eliminate unnecessary rows from our input data
sets as well as an error in the log statement. The following illustration is for the control selection, and the
same code can be substituted for the selection of cases. Note that the output data set from which controls
will be selected is named WORK.ELIGIBLE_CONTROLS. The output data for cases is WORK.
ELIGIBLE_CASES. Also, the data must be sorted by the stratification variable before it goes into PROC
SURVEYSELECT.

PROC SQL;
CREATE TABLE WORK.ELIGIBLE_CONTROLS AS
SELECT *
FROM CTRL
WHERE INDEX IN (SELECT INDEX FROM ALLCOUNT);

PROC SORT DATA = WORK.ELIGIBLE_CONTROLS;
BY INDEX;

SELECT SAMPLES USING PROC SURVEYSELECT
We next employ the SURVEYSELECT procedure separately for controls and cases (using the input data
sets WORK.ELIGIBLE_CONTROLS and WORK.ELIGIBLE_CASES, respectively) to randomly select equal
numbers of patients within each stratum (INDEX) based on the sample size (_NSIZE_) in our look-up table
(WORK.ALLCOUNT).  We optionally choose to indicate the random seed as 499812. The method of
selection is simple random sampling without replacement (METHOD =SRS).  The selected controls are
output to WORK.SELECTED_CONTROLS, while cases are output to WORK.SELECTED_CASES. 

PROC SURVEYSELECT DATA = WORK.ELIGIBLE_CONTROLS 
    SAMPSIZE = ALLCOUNT
    METHOD = SRS 
    SEED=499812
    OUT=WORK.SELECTED_CONTROLS;
    STRATA INDEX;

PROC SURVEYSELECT OUTPUT
Figure 4 displays the output from PROC SURVEYSELECT, which summarizes the sample selection for
controls. A sample of 11 patients is selected using simple random sampling stratified by 5 different strata
values of the variable INDEX. The patient identifiers and their strata are output to a data set of selected
controls named WORK.SELECTED_CONTROLS. Using similar code for cases, the 11 selected cases are
output to WORK.SELECTED_CASES.

Figure 4. The SURVEYSELECT Procedure Output

                           The SURVEYSELECT Procedure

                           Selection Method    Simple Random Sampling
                           Strata Variable     INDEX

                           Input Data Set          ELIGIBLE_CONTROLS
                           Random Number Seed                 499812
                           Sample Size Data Set             ALLCOUNT
                           Number of Strata                        5
                           Total Sample Size                      11
                           Output Data Set         SELECTED_CONTROLS

ADD MATCHING IDENTIFIERS
Once the same number of patients is randomly selected within each stratum, the next step is to identify the
matched pairs.  The two data sets are combined and a case/control identifier, CCID, assigned. 

DATA CC (KEEP=SDYID INDEX CCID);
  SET WORK.SELECTED_CONTROLS (IN=A KEEP=SDYID INDEX) 
      WORK.SELECTED_CASES    (IN=B KEEP=SDYID INDEX);
IF A THEN CCID = 1; *CONTROLS;
ELSE IF B THEN CCID = 0; *CASES;

The following code creates the matched pair identifier, MATCHID, which has identical values for each
matched pair. First sort by the variables INDEX and CCID and create a separate counter for cases and
controls that retains part of the index.  Since the cases and controls were randomly selected within each
index, and the order is random within each index, the resulting counters can be used to create the matched
pair identifier, MATCHID. The final data set, WORK.CC, can be used for a matched pair analysis, Figure 5.

PROC SORT DATA= CC;
BY INDEX CCID;

DATA CC (KEEP=SDYID INDEX CCID MATCHID);
     SET CC;
     BY INDEX CCID;
LENGTH CTKTR CAKTR IDXID  8 IDA $6 MATCHX $50 MATCHID 8;
ATTRIB MATCHID FORMAT =20.;
RETAIN CTKTR CAKTR IDXID;
IF CCID = 1 THEN CTKTR +1; * COUNTER FOR CONTROLS;
ELSE IF CCID = 0 THEN CAKTR +1; * COUNTER FOR CASES;
IF FIRST.INDEX THEN IDXID +1; * INCREASE INDEX COUNT;
IDA = COMPRESS(SUBSTR(INDEX,4,6),'*'); * RETAIN PART OF INDEX;
IDX= PUT(IDXID,$4.); * COUNTER (CHARACTER);
IF CCID = 1 THEN MATCHX = IDX||IDA||CTKTR;     * MATCHID FOR CONTROLS;
ELSE IF CCID = 0 THEN MATCHX = IDX||IDA||CAKTR;* MATCHID FOR CASES;
MATCHX = COMPRESS(MATCHX,'');
MATCHID = INPUT(MATCHX, 20.); * NUMERIC MATCHID;

Figure 5. Final Data set WORK.CC

                        STUDY ID      INDEX      CCID                 MATCHID
                     ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

                     110000793     Bx1025008  CASE                 10250081
                     110000593     Bx1025008  CONTROL              10250081
                     110001763     Bx1025008  CASE                 10250082
                     110000850     Bx1025008  CONTROL              10250082
                     110005242     Bx1037008  CASE                 20370083
                     110001819     Bx1037008  CONTROL              20370083
                     110005252     Cy2043008  CASE                 30430084
                     110005242     Cy2043008  CONTROL              30430084
                     110005288     Cy2043008  CASE                 30430085
                     110005288     Cy2043008  CONTROL              30430085
                     110005759     Cy2043008  CASE                 30430086
                     110005542     Cy2043008  CONTROL              30430086
                     110005832     Cy2043008  CASE                 30430087
                     110005603     Cy2043008  CONTROL              30430087
                     110006324     Cy2043008  CASE                 30430088
                     110005652     Cy2043008  CONTROL              30430088
                     110005361     Cy2043108  CASE                 40431089
                     110006215     Cy2043108  CONTROL              40431089
                     110005841     Cy2043108  CASE                404310810
                     110005022     Cy2043108  CONTROL             404310810
                     110001819     Ggg164008  CASE                516400811
                     110001236     Ggg164008  CONTROL             516400811

CONCLUSION
PROC SURVEYSELECT can be used to simplify the generation of matching cases and controls using
simple random sampling.  This example produces 1:1 matching, but could be customized for 1:N matching
fairly easily. 

There are a host of other options available and at least ten other sampling methods used by this procedure
such as systematic random sampling and probability proportional to size that are not outlined here, but are
worthy of reviewing before selecting your next sample. 

An electronic version of this program can be obtained via e-mail from the author.

